CORRIGE DE CERTAINS EXERCICES DE LA FEUILLE 5

```
reset(); export(linalg);
```

Ecercice 3

```
M:=matrix(4,4,[3,2,2,-4,a,3,2,-1,1,1,2,-1,b,2,2,-1]);
```

$$\begin{pmatrix} 3 & 2 & 2 & -4 \\ a & 3 & 2 & -1 \\ 1 & 1 & 2 & -1 \\ b & 2 & 2 & -1 \end{pmatrix}$$

Expression de a et b sachant que 1 valeur propre double et M diagonalisable

On commence par calculer le polynome caractéristique:

```
MX:=M-x*matrix::identity(4);
```

$$\begin{pmatrix} 3-x & 2 & 2 & -4 \\ a & 3-x & 2 & -1 \\ 1 & 1 & 2-x & -1 \\ b & 2 & 2 & -x-1 \end{pmatrix}$$

```
p:=factor(det(MX));
```

$$(x-1)\cdot(x-3)\cdot\left(-2\cdot a+4\cdot b-3\cdot x+x^2-2\right)$$

prise en compe du fait que 1 est valeur propre double.

Il faut que 1 annule le terme quadratique:

```
h:=expand(p/((x-1)*(x-3))): solve(subs(h,x=1), a);
```

$$\{2\cdot b-2\}$$

Ce qui donne : a = 2(b-1);

M devient :

```
M1:=subs(M,a=2*b-2);
```

$$\begin{pmatrix} 3 & 2 & 2 & -4 \\ 2\cdot b-2 & 3 & 2 & -1 \\ 1 & 1 & 2 & -1 \\ b & 2 & 2 & -1 \end{pmatrix}$$

Et son polynome cractéristique:

```
MX1:= subs(MX,a=2*b-2): p:=factor(det(MX1));
```

$$(x-2)\cdot(x-3)\cdot(x-1)^2$$

Cette écriture montre que 1 est seule valeur propre double.

Il reste à déterminer a pour que le sous-espace propre associé à la valeur propre 1 soit de dim 2.

Pour cela il faut que MX1 pour x=1 soit de rang2


```
MM1:=subs(MX1,x=1);
```

$$\begin{pmatrix} 2 & 2 & 2 & -4 \\ 2\cdot b-2 & 2 & 2 & -1 \\ 1 & 1 & 1 & -1 \\ \dots \end{pmatrix}$$

$$\begin{pmatrix} 2 & 2 & 2 & -4 \\ 2\cdot b-2 & 2 & 2 & -1 \\ 1 & 1 & 1 & -1 \\ b & 2 & 2 & -2 \end{pmatrix}$$

On applique le pivot:

```
pivot:= proc(a,i,j)
local k,m,t;
begin
for k from i+1 to nrows(a) do
t:=a[k,j]/a[i,j]:
for m from j to ncols(a)
do
a[k,m] := a[k,m] - a[i,m] * t:
 end_for:
end_for:
return (a):
end_proc:
```

```
MM2:=pivot(MM1,1,1);
```

$$\begin{pmatrix} 2 & 2 & 2 & -4 \\ 0 & 4-2\cdot b & 4-2\cdot b & 4\cdot b-5 \\ 0 & 0 & 0 & 1 \\ 0 & 2-b & 2-b & 2\cdot b-2 \end{pmatrix}$$

On permute les lignes 3 et 2

```
MM2:=swapRow(MM2,3,2);
```

$$\begin{pmatrix} 2 & 2 & 2 & -4 \\ 0 & 0 & 0 & 1 \\ 0 & 4-2\cdot b & 4-2\cdot b & 4\cdot b-5 \\ 0 & 2-b & 2-b & 2\cdot b-2 \end{pmatrix}$$

Ceci montre que lka matrice est de rang 2 ssi la sou-matrice formée de dernière lignes et 3 première colonne est nulle. En effet

```
MM2:=pivot(MM2,2,4);
```

$$\begin{pmatrix} 2 & 2 & 2 & -4 \\ 0 & 0 & 0 & 1 \\ 0 & 4-2\cdot b & 4-2\cdot b & 0 \\ 0 & 2-b & 2-b & 0 \end{pmatrix}$$

Ce qui montre que la matrice est de rang 2 ssi b=2. et donc a=2.

Vérification

```
M:=subs(M,a=2,b=2);
```

$$\begin{pmatrix} 3 & 2 & 2 & -4 \\ 2 & 3 & 2 & -1 \\ 1 & 1 & 2 & -1 \\ 2 & 2 & 2 & -1 \end{pmatrix}$$

```
J:=jordanForm(M,All);
```


$$\left[\begin{pmatrix} 1 & 0 & 0 & 0 \\ 0 & 2 & 0 & 0 \end{pmatrix}, \begin{pmatrix} -1 & 2 & 0 & \frac{3}{2} \\ 1 & -4 & 3 & -1 \end{pmatrix} \right]$$

$$\left[\begin{pmatrix}1&0&0&0\\0&2&0&0\\0&0&3&0\\0&0&0&1\end{pmatrix},\ \begin{pmatrix}-1&2&0&\frac{3}{2}\\1&-4&3&-1\\0&-1&1&-\frac{1}{2}\\0&-2&2&0\end{pmatrix}\right]$$

Ainsi dans la base formée des colonnes de la matrice P:

```
p:=J[2];
```

$$\begin{pmatrix}-1&2&0&\frac{3}{2}\\1&-4&3&-1\\0&-1&1&-\frac{1}{2}\\0&-2&2&0\end{pmatrix}$$

la matrice M est diagonale:

```
(1/p)* M * p;
```

$$\begin{pmatrix}1&0&0&0\\0&2&0&0\\0&0&3&0\\0&0&0&1\end{pmatrix}$$

Exercice 4

```
reset(); export(linalg);
```

```
A:=matrix([[2,1],[2,3]]); B:=matrix([[0,2],[-2,5]]);
```

$$\begin{pmatrix}2&1\\2&3\end{pmatrix}$$

$$\begin{pmatrix}0&2\\-2&5\end{pmatrix}$$

Montrons qu'elles sont semblables.
Diagonalison ou trigonalisons les:

```
JA:=jordanForm(A,All);JB:=jordanForm(B,All);
```

$$\left[\begin{pmatrix}1&0\\0&4\end{pmatrix},\ \begin{pmatrix}\frac{2}{3}&\frac{1}{3}\\-\frac{2}{3}&\frac{2}{3}\end{pmatrix}\right]$$

$$\left[\begin{pmatrix}1&0\\0&4\end{pmatrix},\ \begin{pmatrix}\frac{4}{3}&-\frac{1}{3}\\\frac{2}{3}&-\frac{2}{3}\end{pmatrix}\right]$$

A et B sont diagonalisables et on les mêmes valeurs propres.
Elles sont semblables. En effet: Leur matrice de passage conduisant à la même forme diagonale est:

```
PA:=JA[2];PB:=JB[2];
```

$$\begin{pmatrix}\frac{2}{3}&\frac{1}{3}\\-\frac{2}{3}&\frac{2}{3}\end{pmatrix}$$

$$\begin{pmatrix}\frac{4}{3}&-\frac{1}{3}\\\frac{2}{3}&-\frac{2}{3}\end{pmatrix}$$

$$\begin{pmatrix} \frac{4}{3} & -\frac{1}{3} \\ \frac{2}{3} & -\frac{2}{3} \end{pmatrix}$$

Soit D la matrice diagonale: on a PA^-1 * A * PA = D = PB^-1 * B * PB
D'où A = PA * PB^-1 * B * PB * PA^-1.
On pose alors C= PB * PA^-1. On aura C^-1 * B * C = A.
Ce qui montre que A et B sont semblables. Enn effet

```
C:= PB * (1/PA);   det(C);   A - (1/C) * B *C;
```

$$\begin{pmatrix} 1 & -1 \\ 0 & -1 \end{pmatrix}$$

$$-1$$

$$\begin{pmatrix} 0 & 0 \\ 0 & 0 \end{pmatrix}$$

A^n et B^n

```
Tn:=matrix([[1,0],[0,4^n]]);
```

$$\begin{pmatrix} 1 & 0 \\ 0 & 4^n \end{pmatrix}$$

```
Bn := PB*(Tn)*(1/PB);
```

$$\begin{pmatrix} \frac{4}{3} - \frac{4^n}{3} & \frac{2 \cdot 4^n}{3} - \frac{2}{3} \\ \frac{2}{3} - \frac{2 \cdot 4^n}{3} & \frac{4 \cdot 4^n}{3} - \frac{1}{3} \end{pmatrix}$$

```
An:=(1/C)*Bn*C;
```

$$\begin{pmatrix} \frac{4^n}{3} + \frac{2}{3} & \frac{4^n}{3} - \frac{1}{3} \\ \frac{2 \cdot 4^n}{3} - \frac{2}{3} & \frac{2 \cdot 4^n}{3} + \frac{1}{3} \end{pmatrix}$$

Exercice 6

```
reset(); export(linalg);
```

```
M:=matrix(3,3,[a,1,0,a+b,0,1-b,2+a ,1,-2]);
```

$$\begin{pmatrix} a & 1 & 0 \\ a+b & 0 & 1-b \\ a+2 & 1 & -2 \end{pmatrix}$$

1/ Polynome caractéristique

```
p:=factor(charpoly(M,x));
```

$$(x+2) \cdot (x+1) \cdot (x-(a+1))$$

Cas particulier où M est susceptible de ne pas être diagonalisable:
C'est le cas où il y aurait des valeur propres multiples


```
solve(a+1 = -2); solve(a+1 = -1)
```

$$\{[a = -3]\}$$

$$\{[a = -2]\}$$

$$\{[a = -2]\}$$

2/ cas a =-3

```
Ma:=subs(M,a=-3);
```

$$\begin{pmatrix} -3 & 1 & 0 \\ b-3 & 0 & 1-b \\ -1 & 1 & -2 \end{pmatrix}$$

```
ja:=jordanForm(Ma,All);
```

$$\left[\begin{pmatrix} -1 & 0 & 0 \\ 0 & -2 & 1 \\ 0 & 0 & -2 \end{pmatrix}, \begin{pmatrix} b-2 & 1-b & 3-b \\ 2\cdot b-4 & 1-b & 4-2\cdot b \\ b-2 & 1-b & 2-b \end{pmatrix}\right]$$

L'instruction nous donne que la matrice est trigonalisable si b est est telle
que la matrice P de passabge est inversible. Cette matrice est

```
Pa:=ja[2];
```

$$\begin{pmatrix} b-2 & 1-b & 3-b \\ 2\cdot b-4 & 1-b & 4-2\cdot b \\ b-2 & 1-b & 2-b \end{pmatrix}$$

calculons son déterminant

```
detpa:=det(Pa);
```

$$-b^2+3\cdot b-2$$

Donc cette matrice n'est pas toujours inversible!!
en effet, pour b dans

```
solve(detpa,b);
```

$$\{1, 2\}$$

cest-àdire b =1 ou b=2, la matrice de passage fournie par Jaordan est n'est pas inversible.
Dans ce cas le résultat fourni par jordanForm est faux. il faut donc étudier ces cas?

2-a/ b=1

```
Mab:=subs(Ma,b=1): jordanForm(Mab,All);
```

$$\left[\begin{pmatrix} -2 & 0 & 0 \\ 0 & -1 & 0 \\ 0 & 0 & -2 \end{pmatrix}, \begin{pmatrix} 2 & -1 & 0 \\ 2 & -2 & 0 \\ 1 & -1 & 1 \end{pmatrix}\right]$$

On voit ici que la matrice est diagonalisable

2-b/ cas b=2

```
Mab:=subs(Ma,b=2): jordanForm(Mab,All);
```

$$\left[\begin{pmatrix} -1 & 0 & 0 \\ 0 & -2 & 1 \\ 0 & 0 & -2 \end{pmatrix}, \begin{pmatrix} -1 & 2 & 0 \\ -2 & 2 & 2 \\ -1 & 2 & 2 \end{pmatrix}\right]$$

Ici Par contre la matrice est seulement trigonalisable mais pas diagonalisable!!!!

3/ L'autre cas a:=-2

```
Ma:=subs(M,a=-2);
```


$$\begin{pmatrix} -2 & 1 & 0 \\ b-2 & 0 & 1-b \\ 0 & 1 & -2 \end{pmatrix}$$

$$\begin{pmatrix} -2 & 1 & 0 \\ b-2 & 0 & 1-b \\ 0 & 1 & -2 \end{pmatrix}$$

```
ja:=jordanForm(Ma,All);
```

$$\left[\begin{pmatrix} -2 & 0 & 0 \\ 0 & -1 & 1 \\ 0 & 0 & -1 \end{pmatrix}, \begin{pmatrix} b-1 & b-2 & 2-b \\ 0 & b-2 & 0 \\ b-2 & b-2 & 2-b \end{pmatrix}\right]$$

```
Pa:=ja[2];
```

$$\begin{pmatrix} b-1 & b-2 & 2-b \\ 0 & b-2 & 0 \\ b-2 & b-2 & 2-b \end{pmatrix}$$

```
solve(det(Pa),b);
```

$$\{2\}$$

on trouve qu'il y a seulement une valeur de b =1. On tire comme conclusion que si b!=2, alors la matrice est seulement trigonalisable.

Etudions le cas b=2

```
Mab:=subs(Ma,b=2): jordanForm(Mab,All);
```

$$\left[\begin{pmatrix} -2 & 0 & 0 \\ 0 & -1 & 1 \\ 0 & 0 & -1 \end{pmatrix}, \begin{pmatrix} -1 & -1 & 1 \\ 0 & -1 & 0 \\ 0 & -1 & 1 \end{pmatrix}\right]$$

On remarque que même pour b=2 la matrice n'est pas diagonalisable.
Elle est seulement trigonalisable

4/ a =0, b=0. Calcul de Mn = M^n

```
M1:=subs(M,a=0,b=0);
```

$$\begin{pmatrix} 0 & 1 & 0 \\ 0 & 0 & 1 \\ 2 & 1 & -2 \end{pmatrix}$$

diagonalisation ou trigonalisation

```
jm:=jordanForm(M1,All);
```

$$\left[\begin{pmatrix} -2 & 0 & 0 \\ 0 & -1 & 0 \\ 0 & 0 & 1 \end{pmatrix}, \begin{pmatrix} -\frac{1}{3} & 1 & \frac{1}{3} \\ \frac{2}{3} & -1 & \frac{1}{3} \\ -\frac{4}{3} & 1 & \frac{1}{3} \end{pmatrix}\right]$$

On trouve A diagonalisable avec pour matrice de passage

```
pM1:=jm[2];
```

$$\begin{pmatrix} -\frac{1}{3} & 1 & \frac{1}{3} \\ \frac{2}{3} & -1 & \frac{1}{3} \\ -\frac{4}{3} & 1 & \frac{1}{3} \end{pmatrix}$$

et la matrice diagonale

et la matrice diagonale

```
dM1:=jm[1];
```

$$\begin{pmatrix} -2 & 0 & 0 \\ 0 & -1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

On calcul dM^n

```
dMn:=matrix(3,3,[(-2)^n, (-1)^n,1],Diagonal)
```

$$\begin{pmatrix} (-2)^n & 0 & 0 \\ 0 & (-1)^n & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

On en déduit Mn

```
Mn:= pM1 * dMn * (1/pM1);
```

$$\begin{pmatrix} (-1)^n - \frac{(-2)^n}{3} + \frac{1}{3} & \frac{1}{2} - \frac{(-1)^n}{2} & \frac{(-2)^n}{3} - \frac{(-1)^n}{2} + \frac{1}{6} \\ \frac{2\cdot(-2)^n}{3} - (-1)^n + \frac{1}{3} & \frac{(-1)^n}{2} + \frac{1}{2} & \frac{(-1)^n}{2} - \frac{2\cdot(-2)^n}{3} + \frac{1}{6} \\ (-1)^n - \frac{4\cdot(-2)^n}{3} + \frac{1}{3} & \frac{1}{2} - \frac{(-1)^n}{2} & \frac{4\cdot(-2)^n}{3} - \frac{(-1)^n}{2} + \frac{1}{6} \end{pmatrix}$$

4/ a =-3 b=2. Calcul de Mn = M^n

```
M2:=subs(M,a=-3,b=2);
```

$$\begin{pmatrix} -3 & 1 & 0 \\ -1 & 0 & -1 \\ -1 & 1 & -2 \end{pmatrix}$$

```
jm:=jordanForm(M2,All);
```

$$\left[\begin{pmatrix} -1 & 0 & 0 \\ 0 & -2 & 1 \\ 0 & 0 & -2 \end{pmatrix}, \begin{pmatrix} -1 & 2 & 0 \\ -2 & 2 & 2 \\ -1 & 2 & 2 \end{pmatrix}\right]$$

Ici M2 n'est pas diagonalisable

```
T:=jm[1]; P:=jm[2];
```

$$\begin{pmatrix} -1 & 0 & 0 \\ 0 & -2 & 1 \\ 0 & 0 & -2 \end{pmatrix}$$

$$\begin{pmatrix} -1 & 2 & 0 \\ -2 & 2 & 2 \\ -1 & 2 & 2 \end{pmatrix}$$

On deécompose T est partie diagonale et sa partie nilpotente

```
T1:=matrix(3,3,i->T[i,i],Diagonal);  N:= T-T1;
```

$$\begin{pmatrix} -1 & 0 & 0 \\ 0 & -2 & 0 \\ 0 & 0 & -2 \end{pmatrix}$$

$$\begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 1 \\ 0 & 0 & 0 \end{pmatrix}$$

On vérifie que les deux commutent:

```
N*T1 - T1*N;
```

$$\begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

$$\begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

On cherche le degré de nilpotence de N

```
N1:=N;
```

$$\begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 1 \\ 0 & 0 & 0 \end{pmatrix}$$

```
N2:=N*N;
```

$$\begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

le degré de nilpotence de N est 2.
On utilise la formule du binome de Newton (voir cours) pour calculer T^n.
on sait que l'on va s'arrêter à 2.
On note : T1n=T1^n et T1n1 = T1^(n-1).
Qu'on calcul simplement

```
T1n:= matrix(3,3,i->(T[i,i])^n,Diagonal);
T1n1:= matrix(3,3,i->(T[i,i])^(n-1),Diagonal);
```

$$\begin{pmatrix} (-1)^n & 0 & 0 \\ 0 & (-2)^n & 0 \\ 0 & 0 & (-2)^n \end{pmatrix}$$

$$\begin{pmatrix} (-1)^{n-1} & 0 & 0 \\ 0 & (-2)^{n-1} & 0 \\ 0 & 0 & (-2)^{n-1} \end{pmatrix}$$

Et finalement

```
Tn := T1n + n * T1n1 * N;
```

$$\begin{pmatrix} (-1)^n & 0 & 0 \\ 0 & (-2)^n & (-2)^{n-1} \cdot n \\ 0 & 0 & (-2)^n \end{pmatrix}$$

On en déduit Mn

```
Mn:=P * Tn * (1/P);
```

$$\begin{pmatrix} (-2)^n - (-2)^{n-1} \cdot n & (-1)^n - (-2)^n & (-2)^n - (-1)^n + (-2)^{n-1} \cdot n \\ -(-2)^{n-1} \cdot n & 2 \cdot (-1)^n - (-2)^n & 2 \cdot (-2)^n - 2 \cdot (-1)^n + (-2)^{n-1} \cdot n \\ -(-2)^{n-1} \cdot n & (-1)^n - (-2)^n & 2 \cdot (-2)^n - (-1)^n + (-2)^{n-1} \cdot n \end{pmatrix}$$

Exercice 7


```
reset(); export(linalg);
```

```
A:=(1/2)* matrix([[7,5,-13],[-3,-1,9],[1,1,0]]);
```

$$\begin{pmatrix} \frac{7}{2} & \frac{5}{2} & -\frac{13}{2} \\ -\frac{3}{2} & -\frac{1}{2} & \frac{9}{2} \\ \frac{1}{2} & \frac{1}{2} & 0 \end{pmatrix}$$

1/ forme diagonale ou triagulaire de A et calcul de A^n

```
ja:=jordanForm(A,All);
```

$$\left[\begin{pmatrix} 1 & 1 & 0 \\ 0 & 1 & 1 \\ 0 & 0 & 1 \end{pmatrix}, \begin{pmatrix} -\frac{3}{4} & \frac{5}{2} & 1 \\ \frac{3}{4} & -\frac{3}{2} & 0 \\ 0 & \frac{1}{2} & 0 \end{pmatrix} \right]$$

A n'est pas diagonalisable

```
T:=ja[1]; P:=ja[2];
```

$$\begin{pmatrix} 1 & 1 & 0 \\ 0 & 1 & 1 \\ 0 & 0 & 1 \end{pmatrix}$$

$$\begin{pmatrix} -\frac{3}{4} & \frac{5}{2} & 1 \\ \frac{3}{4} & -\frac{3}{2} & 0 \\ 0 & \frac{1}{2} & 0 \end{pmatrix}$$

Calcul de Tn = T^n;

```
T1:=matrix(3,3,i->T[i,i],Diagonal); N:=T-T1;
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

$$\begin{pmatrix} 0 & 1 & 0 \\ 0 & 0 & 1 \\ 0 & 0 & 0 \end{pmatrix}$$

Order de ninpotence de N

```
N1:=N; N2:=N1*N; N3:=N2*N;
```

$$\begin{pmatrix} 0 & 1 & 0 \\ 0 & 0 & 1 \\ 0 & 0 & 0 \end{pmatrix}$$

$$\begin{pmatrix} 0 & 0 & 1 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

$$\begin{pmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{pmatrix}$$

On voit que l'order de nilpotence de N est 3.
d'où:
T^n = T1^n + n * T1^(n-1)*N + (n!)/((n-2)! *(2!)) * T1^(n-2)*N^2
On peut tout simplement remarquer que T1 est la matrice identité.
Le calcul ci-dessous es donc fourni pour être utilisé dans un cadre général.

```
T1n:=matrix(3,3,i->(T[i,i])^n,Diagonal);
T1n1:=matrix(3,3,i->(T[i,i])^(n-1),Diagonal);
T1n2:=matrix(3,3,i->(T[i,i])^(n-2),Diagonal);
```


$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

On calcul alors Tn

```
Tn:= T1n + n *T1n1 * N +  expand( (n!)/( (n-2)!  *(2!) )) * T1n2 * N2;
```

$$\begin{pmatrix} 1 & n & \frac{n^2}{2} - \frac{n}{2} \\ 0 & 1 & n \\ 0 & 0 & 1 \end{pmatrix}$$

D'où l'on déduit Mn

```
Mn:= P * Tn *(1/P);
```

$$\begin{pmatrix} \frac{23 \cdot n}{8} - \frac{3 \cdot n^2}{8} + 1 & \frac{23 \cdot n}{8} - \frac{3 \cdot n^2}{8} & \frac{3 \cdot n^2}{4} - \frac{29 \cdot n}{4} \\ \frac{3 \cdot n^2}{8} - \frac{15 \cdot n}{8} & \frac{3 \cdot n^2}{8} - \frac{15 \cdot n}{8} + 1 & \frac{21 \cdot n}{4} - \frac{3 \cdot n^2}{4} \\ \frac{n}{2} & \frac{n}{2} & 1 - n \end{pmatrix}$$

Application à l'étude de suite:

On remarque que si l'on pose Xn =matrix[[u_n,v_n,w_n]]) , on a X_{n+1} = M X_n.

Ainsi X_n = M^n X_0

```
X0:=matrix([u0,v0,w0]);
```

$$\begin{pmatrix} u0 \\ v0 \\ w0 \end{pmatrix}$$

```
Xn:= Mn * X0;
```

$$\begin{pmatrix} u0 \cdot \left(\frac{23 \cdot n}{8} - \frac{3 \cdot n^2}{8} + 1\right) + v0 \cdot \left(\frac{23 \cdot n}{8} - \frac{3 \cdot n^2}{8}\right) - w0 \cdot \left(\frac{29 \cdot n}{4} - \frac{3 \cdot n^2}{4}\right) \\ v0 \cdot \left(\frac{3 \cdot n^2}{8} - \frac{15 \cdot n}{8} + 1\right) - u0 \cdot \left(\frac{15 \cdot n}{8} - \frac{3 \cdot n^2}{8}\right) + w0 \cdot \left(\frac{21 \cdot n}{4} - \frac{3 \cdot n^2}{4}\right) \\ \frac{n \cdot u0}{2} - w0 \cdot (n - 1) + \frac{n \cdot v0}{2} \end{pmatrix}$$

d'où

```
un:= Xn[1];
vn:= Xn[2];
wn:= Xn[3];
```

$$u0 \cdot \left(-\frac{3 \cdot n^2}{8} + \frac{23 \cdot n}{8} + 1\right) + v0 \cdot \left(\frac{23 \cdot n}{8} - \frac{3 \cdot n^2}{8}\right) - w0 \cdot \left(\frac{29 \cdot n}{4} - \frac{3 \cdot n^2}{4}\right)$$

$$v0 \cdot \left(\frac{3 \cdot n^2}{8} - \frac{15 \cdot n}{8} + 1\right) - u0 \cdot \left(\frac{15 \cdot n}{8} - \frac{3 \cdot n^2}{8}\right) + w0 \cdot \left(\frac{21 \cdot n}{4} - \frac{3 \cdot n^2}{4}\right)$$

$$\frac{n \cdot u0}{2} - w0 \cdot (n - 1) + \frac{n \cdot v0}{2}$$

ce sont des polynomes en n:

```
un:= collect(Xn[1],n);
vn:= collect(Xn[2],n);
```

```
vn:= collect(Xn[2],n);
wn:= collect(Xn[3],n);
```

$$\left(\frac{3\cdot w0}{4}-\frac{3\cdot v0}{8}-\frac{3\cdot u0}{8}\right)\cdot n^2+\left(\frac{23\cdot u0}{8}+\frac{23\cdot v0}{8}-\frac{29\cdot w0}{4}\right)\cdot n+u0$$

$$\left(\frac{3\cdot u0}{8}+\frac{3\cdot v0}{8}-\frac{3\cdot w0}{4}\right)\cdot n^2+\left(\frac{21\cdot w0}{4}-\frac{15\cdot v0}{8}-\frac{15\cdot u0}{8}\right)\cdot n+v0$$

$$\left(\frac{u0}{2}+\frac{v0}{2}-w0\right)\cdot n+w0$$

puisque ce sont des polynomes, il y aura des limites finies si elles sont constantes.
Formons l'équation:

```
eq:=[coeff(un,2), coeff(un,1), coeff(vn,2), coeff(vn,1),coeff(wn,1)];
```

$$\left[\frac{3\cdot w0}{4}-\frac{3\cdot v0}{8}-\frac{3\cdot u0}{8},\ \frac{23\cdot u0}{8}+\frac{23\cdot v0}{8}-\frac{29\cdot w0}{4},\ \frac{3\cdot u0}{8}+\frac{3\cdot v0}{8}-\frac{3\cdot w0}{4},\ \frac{21\cdot w0}{4}-\frac{15\cdot v0}{8}$$

Qu'on résout

```
solve(eq,[u0,v0,w0]);
```

$$\{[u0=-z,\ v0=z,\ w0=0]\}$$

On trouve que u0 = v0 et W0=0 pour qu'il y'ait une limite

C/ Condition sur u0,v0,w0 pourque wn ait une limite finie.
Cette condition est

```
coeff(wn,1) = 0;
```

$$\frac{u0}{2}+\frac{v0}{2}-w0=0$$

soit w0 = (u0+v0)/2.
Les nouvelle expression de un et vn sont

```
un:=subs(un,w0 = (u0+v0)/2);
vn:=subs(vn,w0 = (u0+v0)/2);
```

$$u0-n\cdot\left(\frac{3\cdot u0}{4}+\frac{3\cdot v0}{4}\right)$$

$$v0+n\cdot\left(\frac{3\cdot u0}{4}+\frac{3\cdot v0}{4}\right)$$

On remarque que

```
un + vn
```

$$u0+v0$$

donc un +vn = cte donc un ~ -vn. c'est -à-dire que un et (-vn) sont équivalentes.